
  🔱 of XmlBuilder


This is a fork of XmlBuilder originally written by Joshua Nussbaum. OIt’s fully backward compatible and adds several goodnesses like Access implementation for XML‌ AST, custom formatters, per-element formatters, listeners and pre-/post-walk of the resulting AST.
If in doubt, why would you want to use this fork, please use the original XmlBuilder package, it’d suit your needs perfectly fine.

  
    
  
  Overview


An Elixir library for building XML. It is inspired by the late Jim Weirich's awesome builder library for Ruby.
Each XML node is structured as a tuple of name, attributes map, and content/list.
{name, attrs, content | list}

  
    
  
  Installation


Add dependency to your project's mix.exs:
def deps do
  [{:xml_builder_ex, "~> 3.1"}]
end

  
    
  
  Examples



  
    
  
  A simple element


Like <person id="12345">Josh</person>, would look like:
{:person, %{id: 12345}, "Josh"} |> XmlBuilder.generate()

  
    
  
  An element with child elements


Like <person id="12345"><first>Josh</first><last>Nussbaum</last></person>.
{:person, %{id: 12345}, [{:first, nil, "Josh"}, {:last, nil, "Nussbaum"}]} |> XmlBuilder.generate()

  
    
  
  Convenience Functions


For more readability, you can use XmlBuilder's methods instead of creating tuples manually.
XmlBuilder.document(:person, "Josh") |> XmlBuilder.generate()
Outputs:
<?xml version="1.0" encoding="UTF-8" ?>
<person>Josh</person>
Building up an element
An element can be built using multiple calls to the element function.
import XmlBuilder

def person(id, first, last) do
  element(:person, %{id: id}, [
    element(:first, first),
    element(:last, last)
  ])
end

iex> [person(123, "Steve", "Jobs"),
      person(456, "Steve", "Wozniak")] |> generate()
Outputs.
<person id="123">
  <first>Steve</first>
  <last>Jobs</last>
</person>
<person id="456">
  <first>Steve</first>
  <last>Wozniak</last>
</person>
Using keyed lists
The previous example can be simplified using a keyed list.
import XmlBuilder

def person(id, first, last) do
  element(:person, %{id: id}, first: first,
                              last: last)
end

iex> person(123, "Josh", "Nussbaum") |> generate(format: :none)
"<person id=\"123\"><first>Josh</first><last>Nussbaum</last></person>"
Namespaces
To use a namespace, add an xmlns attribute to the root element.
To use multiple schemas, specify a xmlns:nsName attribute for each schema and use a colon : in the element name, ie nsName:elementName.
import XmlBuilder

iex> generate({:example, [xmlns: "http://schemas.example.tld/1999"], "content"})
"<example xmlns=\"http://schemas.example.tld/1999\">content</example>"

iex> generate({:"nsName:elementName", ["xmlns:nsName": "http://schemas.example.tld/1999"], "content"})
"<nsName:elementName xmlns:nsName=\"http://schemas.example.tld/1999\">content</nsName:elementName>"

  
    
  
  DOCTYPE declarations


A DOCTYPE can be declared by applying the doctype function at the first position of a list of elements in a document definition:
import XmlBuilder

document([
  doctype("html", public: ["-//W3C//DTD XHTML 1.0 Transitional//EN",
                "http://www.w3.org/TR/xhtml1/DTD/xhtml1-transitional.dtd"]),
  element(:html, "Hello, world!")
]) |> generate()
Outputs.
<?xml version="1.0" encoding="UTF-8"?>
<!DOCTYPE html PUBLIC "-//W3C//DTD XHTML 1.0 Transitional//EN" "http://www.w3.org/TR/xhtml1/DTD/xhtml1-transitional.dtd">
<html>Hello, world!</html>

  
    
  
  Encoding


While the output is always UTF-8 and has to be converted in another place, you can override the encoding statement in the XML declaration with the encoding option.
import XmlBuilder

document(:oldschool)
|> generate(encoding: "ISO-8859-1")
|> :unicode.characters_to_binary(:unicode, :latin1)
Outputs.
<?xml version="1.0" encoding="ISO-8859-1"?>
<oldschool/>

  
    
  
  Using iodata() directly


While by default, output from generate/2 is converted to binary(), you can use generate_iodata/2 to skip this conversion. This can be convenient if you're using IO.binwrite/2 on a :raw IO device, as these APIs can work with iodata() directly, leading to some performance gains.
In some scenarios, it may be beneficial to generate part of your XML upfront, for instance when generating a sitemap.xml, you may have shared fields for author. Instead of generating this each time, you could do the following:
import XmlBuilder

entries = [%{title: "Test", url: "https://example.org/"}]

# Generate static author data upfront
author = generate_iodata(element(:author, [
  element(:name, "John Doe"),
  element(:uri, "https://example.org/")
]))

file = File.open!("path/to/file", [:raw])

for entry <- entries do
  iodata =
    generate_iodata(element(:entry, [
      # Reuse the static pre-generated fields as-is
      {:iodata, author},

      # Dynamic elements are generated for each entry
      element(:title, entry.title),
      element(:link, entry.url)
    ]))

  IO.binwrite(file, iodata)
end

  
    
  
  Escaping


XmlBuilder offers 3 distinct ways to control how content of tags is escaped and handled:
	By default, any content is escaped, replacing reserved characters (& " ' < >) with their equivalent entity (&amp; etc.)
	If content is wrapped in {:cdata, cdata}, the content in cdata is wrapped with <![CDATA[...]]>, and not escaped. You should make sure the content itself does not contain ]]>.
	If content is wrapped in {:safe, data}, the content in data is not escaped, but will be stringified if not a bitstring. Use this option carefully. It may be useful when data is guaranteed to be safe (numeric data).
	If content is wrapped in {:iodata, data}, either in the top level or within a list, the data is used as iodata(), and will not be escaped, indented or stringified. An example of this can be seen in the "Using iodata() directly" example above.


  
    
  
  Standalone


Should you need standalone="yes" in the XML declaration, you can pass standalone: true as option to the generate/2 call.
import XmlBuilder

:outsider
|> document()
|> generate(standalone: true)
Outputs.
<?xml version="1.0" standalone="yes"?>
<outsider/>
If otherwise you need standalone ="no" in the XML declaration, you can pass standalone: false as an option to thegenerate / 2 call.
Outputs.
<?xml version="1.0" standalone="no"?>
<outsider/>

  
    
  
  Formatting


To remove indentation, pass format: :none option to XmlBuilder.generate/2.
doc |> XmlBuilder.generate(format: :none)
The default is to formatting with indentation, which is equivalent to XmlBuilder.generate(doc, format: :indent).

  
    
  
  Access


Use xml_builder_access library to get the Access implemented for
the generated XML tree.
  {:xml_builder_access, "~> 1.0"}
	github
	docs


  
    
  
  Changelog


	3.1.41 Application.get_env(:xml_builder, :arrogant_floats) and Application.get_env(:xml_builder, :float_notation)
	3.1.4 empty: :squeezed option in a call to XmlBuilder.generate/2
	3.1.3 empty: :full option in a call to XmlBuilder.generate/2
	3.1.0 XmlBuilder.{traverse,prewalk,postwalk}
	3.0.0 initial fork of the original repo
	3.0.1 formats, refactoring, specs
	3.1.0 XmlBuilder.{pre,post}walk/{2,3}


  
    
  
  License


This source code is licensed under the MIT License. Copyright (c) 2014-present, Joshua Nussbaum. All rights reserved.



  

    
XmlBuilder 
    



      
A module for generating XML

  
    
  
  Examples


iex> XmlBuilder.document(:person) |> XmlBuilder.generate()
"<?xml version=\"1.0\" encoding=\"UTF-8\"?>\n<person/>"

iex> XmlBuilder.document(:person, "Josh") |> XmlBuilder.generate()
"<?xml version=\"1.0\" encoding=\"UTF-8\"?>\n<person>Josh</person>"

iex> XmlBuilder.document(:person) |> XmlBuilder.generate(format: :none)
"<?xml version=\"1.0\" encoding=\"UTF-8\"?><person/>"

iex> XmlBuilder.element(:person, "Josh") |> XmlBuilder.generate()
"<person>Josh</person>"

iex> XmlBuilder.element(:person, %{occupation: "Developer"}, "Josh") |> XmlBuilder.generate()
"<person occupation=\"Developer\">Josh</person>"

      


      
        Summary


  
    Functions
  


    
      
        doctype(name, list)

      


        Creates a DOCTYPE declaration with a system or public identifier.



    


    
      
        document(elements)

      


        Generate an XML document.



    


    
      
        document(name, attrs_or_content)

      


    


    
      
        document(name, attrs, content)

      


    


    
      
        element(a1)

      


        Create an XML element.



    


    
      
        element(a1, a2)

      


        See XmlBuilder.Element.element/2.



    


    
      
        element(a1, a2, a3)

      


        See XmlBuilder.Element.element/3.



    


    
      
        generate(any, opts \\ [])

      


        Generate a binary from an XML tree. Accepts optional parameters



    


    
      
        generate_iodata(any, opts \\ [])

      


        Similar to generate/2, but returns iodata instead of a binary.



    


    
      
        postwalk(element, post)

      


        Performs a depth-first, post-order traversal of XML tree.



    


    
      
        postwalk(element, acc, post)

      


        Performs a depth-first, post-order traversal of XML tree.



    


    
      
        prewalk(element, pre)

      


        Performs a depth-first, pre-order traversal of XML tree.



    


    
      
        prewalk(element, acc, pre)

      


        Performs a depth-first, pre-order traversal of XML tree.



    


    
      
        traverse(element, acc, pre, post)

      


        Performs a depth-first traversal of XML tree using an accumulator.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    doctype(name, list)


      
       
       View Source
     


  


  

      

          @spec doctype(XmlBuilder.Element.name(), system: binary(), public: [binary()]) ::
  {:doctype,
   {:system, XmlBuilder.Element.name(), binary()}
   | {:public, XmlBuilder.Element.name(), binary(), binary()}}


      


Creates a DOCTYPE declaration with a system or public identifier.

  
    
  
  System Example


Returns a tuple in the format {:doctype, {:system, name, system_identifier}}.
import XmlBuilder

document([
  doctype("greeting", system: "hello.dtd"),
  element(:person, "Josh")
]) |> generate()
Outputs
<?xml version="1.0" encoding="UTF-8" ?>
<!DOCTYPE greeting SYSTEM "hello.dtd">
<person>Josh</person>

  
    
  
  Public Example


 Returns a tuple in the format {:doctype, {:public, name, public_identifier, system_identifier}}.
import XmlBuilder

document([
  doctype("html", public: ["-//W3C//DTD XHTML 1.0 Transitional//EN",
                "http://www.w3.org/TR/xhtml1/DTD/xhtml1-transitional.dtd"]),
  element(:html, "Hello, world!")
]) |> generate()
Outputs
<?xml version="1.0" encoding="UTF-8"?>
<!DOCTYPE html PUBLIC "-//W3C//DTD XHTML 1.0 Transitional//EN" "http://www.w3.org/TR/xhtml1/DTD/xhtml1-transitional.dtd">
<html>Hello, world!</html>

  



  
    
      
      Link to this function
    
    document(elements)


      
       
       View Source
     


  


  

      

          @spec document(XmlBuilder.Element.ast()) :: XmlBuilder.Element.ast()


      


Generate an XML document.
Returns a binary.

  
    
  
  Examples


iex> XmlBuilder.document(:person) |> XmlBuilder.generate()
"<?xml version=\"1.0\" encoding=\"UTF-8\"?>\n<person/>"

iex> XmlBuilder.document(:person, %{id: 1}) |> XmlBuilder.generate()
"<?xml version=\"1.0\" encoding=\"UTF-8\"?>\n<person id=\"1\"/>"

iex> XmlBuilder.document(:person, %{id: 1}, "some data") |> XmlBuilder.generate()
"<?xml version=\"1.0\" encoding=\"UTF-8\"?>\n<person id=\"1\">some data</person>"

  



  
    
      
      Link to this function
    
    document(name, attrs_or_content)


      
       
       View Source
     


  


  

      

          @spec document(
  XmlBuilder.Element.name(),
  XmlBuilder.Element.attrs() | XmlBuilder.Element.contents()
) ::
  XmlBuilder.Element.ast()


      



  



  
    
      
      Link to this function
    
    document(name, attrs, content)


      
       
       View Source
     


  


  

      

          @spec document(
  XmlBuilder.Element.name(),
  XmlBuilder.Element.attrs(),
  XmlBuilder.Element.contents()
) ::
  XmlBuilder.Element.ast()


      



  



  
    
      
      Link to this function
    
    element(a1)


      
       
       View Source
     


  


  

Create an XML element.
Returns a tuple in the format {name, attributes, content | list}.

  
    
  
  Examples


iex> XmlBuilder.element(:person)
{:person, nil, nil}

iex> XmlBuilder.element(:person, "data")
{:person, nil, "data"}

iex> XmlBuilder.element(:person, %{id: 1})
{:person, %{id: 1}, nil}

iex> XmlBuilder.element(:person, %{id: 1}, "data")
{:person, %{id: 1}, "data"}

iex> XmlBuilder.element(:person, %{id: 1}, [XmlBuilder.element(:first, "Steve"), XmlBuilder.element(:last, "Jobs")])
{:person, %{id: 1}, [
  {:first, nil, "Steve"},
  {:last, nil, "Jobs"}
]}

  



  
    
      
      Link to this function
    
    element(a1, a2)


      
       
       View Source
     


  


  

See XmlBuilder.Element.element/2.

  



  
    
      
      Link to this function
    
    element(a1, a2, a3)


      
       
       View Source
     


  


  

See XmlBuilder.Element.element/3.

  



    

  
    
      
      Link to this function
    
    generate(any, opts \\ [])


      
       
       View Source
     


  


  

      

          @spec generate(
  XmlBuilder.Element.ast(),
  keyword()
) :: String.t()


      


Generate a binary from an XML tree. Accepts optional parameters
	format: Format.Module.Name to specify the formatter to use,
the format parameter might be shortened to :none and :indented for built-in formatters
	empty: :full | :squeezed to not shorten empty tags <foo></foo> to <foo/>,
:full will preserve the indentation, :squeezed would produce closing tag immediately after
the opening one, without spaces


Returns a binary.

  
    
  
  Examples


iex> XmlBuilder.generate(XmlBuilder.element(:person))
"<person/>"

iex> XmlBuilder.generate({:person, %{id: 1}, "Steve Jobs"})
"<person id=\"1\">Steve Jobs</person>"

iex> XmlBuilder.generate({:name, nil, [{:first, nil, "Steve"}]}, format: :none)
"<name><first>Steve</first></name>"

iex> XmlBuilder.generate({:name, nil, [{:first, nil, "Steve"}]})
"<name>\n  <first>Steve</first>\n</name>"

iex> XmlBuilder.generate(:xml_decl, encoding: "ISO-8859-1")
~s|<?xml version="1.0" encoding="ISO-8859-1"?>|

iex> XmlBuilder.generate(
...>  [{:person, %{},
...>    [{:name, %{id: 123}, "Josh"},
...>     {:age, %{}, "21"}]}], format: XmlBuilder.Format.None)
"<person><name id=\"123\">Josh</name><age>21</age></person>"

iex> XmlBuilder.generate(
...>  [{:person, %{},
...>    [{:name, %{id: 123}, "Josh"},
...>     {:age, %{}, "21"}]}], format: :none)
"<person><name id=\"123\">Josh</name><age>21</age></person>"

  



    

  
    
      
      Link to this function
    
    generate_iodata(any, opts \\ [])


      
       
       View Source
     


  


  

      

          @spec generate_iodata(
  XmlBuilder.Element.ast(),
  keyword()
) :: iodata()


      


Similar to generate/2, but returns iodata instead of a binary.

  
    
  
  Examples


iex> XmlBuilder.generate_iodata(XmlBuilder.element(:person))
["", '<', "person", '/>']

  



  
    
      
      Link to this function
    
    postwalk(element, post)


      
       
       View Source
     


  


  

      

          @spec postwalk(
  XmlBuilder.Element.ast(),
  (XmlBuilder.Element.as_tuple() -> XmlBuilder.Element.as_tuple())
) :: XmlBuilder.Element.as_tuple()


      


Performs a depth-first, post-order traversal of XML tree.
Returns a new tree of XmlBuilder.Element.t() where each element
  is the result of invoking fun on each corresponding element.

  
    
  
  Examples


iex> XmlBuilder.postwalk(XmlBuilder.element(:person), fn
...>   {x, nil, nil} -> {x, %{}, []}
...> end)
{:person, %{}, []}

iex> XmlBuilder.postwalk({:person, %{id: 1}, "Steve Jobs"}, fn
...>   {name, args, _content} -> {name, args, "Linux Torvalds"}
...> end)
{:person, %{id: 1}, "Linux Torvalds"}

iex> XmlBuilder.postwalk({:name, nil, [{:first, nil, "Steve"}]}, fn
...>   {:first, _, _} -> {:first, %{}, "Linus"}
...>   e -> e
...> end)
{:name, nil, [{:first, %{}, "Linus"}]}

iex> XmlBuilder.postwalk(
...>  [{:person, %{},
...>    [{:name, %{id: 123}, "Josh"},
...>     {:age, %{}, "21"}]}], fn
...>   {:age, _, _} -> {:age, %{}, "24"}
...>   e -> e
...> end)
...> |> hd()
...> |> XmlBuilder.generate(format: XmlBuilder.Format.None)
~s|<person><name id="123">Josh</name><age>24</age></person>|

  



  
    
      
      Link to this function
    
    postwalk(element, acc, post)


      
       
       View Source
     


  


  

      

          @spec postwalk(
  XmlBuilder.Element.ast(),
  any(),
  (XmlBuilder.Element.as_tuple(), any() ->
     {XmlBuilder.Element.as_tuple(), any()})
) :: {XmlBuilder.Element.as_tuple(), any()}


      


Performs a depth-first, post-order traversal of XML tree.
Returns a new tree of XmlBuilder.Element.t() where each element
  is the result of invoking fun on each corresponding element.

  
    
  
  Examples


iex> XmlBuilder.postwalk(XmlBuilder.element(:person), %{}, fn
...>   {x, nil, nil}, a -> {{x, %{}, []}, a}
...> end)
{{:person, %{}, []}, %{}}

iex> XmlBuilder.postwalk({:person, %{id: 1}, "Steve Jobs"}, %{}, fn
...>   {name, args, _content}, a -> {{name, args, "Linux Torvalds"}, a}
...> end)
{{:person, %{id: 1}, "Linux Torvalds"}, %{}}

iex> XmlBuilder.postwalk({:name, nil, [{:first, nil, "Steve"}]}, %{}, fn
...>   {:first, _, value}, a ->
...>     {{:first, %{}, "Linus"}, Map.put(a, :first, value)}
...>   e, a ->
...>     {e, a}
...> end)
{{:name, nil, [{:first, %{}, "Linus"}]}, %{first: "Steve"}}

iex> XmlBuilder.postwalk(
...>  [{:person, %{},
...>    [{:name, %{id: 123}, "Josh"},
...>     {:age, %{}, "21"}]}], %{}, fn
...>   {:age, _, _}, a -> {{:age, %{}, "24"}, a}
...>   e, a -> {e, a}
...> end)
...> |> elem(0)
...> |> hd()
...> |> XmlBuilder.generate(format: XmlBuilder.Format.None)
~s|<person><name id="123">Josh</name><age>24</age></person>|

  



  
    
      
      Link to this function
    
    prewalk(element, pre)


      
       
       View Source
     


  


  

      

          @spec prewalk(
  XmlBuilder.Element.ast(),
  (XmlBuilder.Element.as_tuple() -> XmlBuilder.Element.as_tuple())
) :: XmlBuilder.Element.as_tuple()


      


Performs a depth-first, pre-order traversal of XML tree.
Returns a new tree of XmlBuilder.Element.t() where each element
  is the result of invoking fun on each corresponding element.

  
    
  
  Examples


iex> XmlBuilder.prewalk(XmlBuilder.element(:person), fn
...>   {x, nil, nil} -> {x, %{}, []}
...> end)
{:person, %{}, []}

iex> XmlBuilder.prewalk({:person, %{id: 1}, "Steve Jobs"}, fn
...>   {name, args, _content} -> {name, args, "Linux Torvalds"}
...> end)
{:person, %{id: 1}, "Linux Torvalds"}

iex> XmlBuilder.prewalk({:name, nil, [{:first, nil, "Steve"}]}, fn
...>   {:first, _, _} -> {:first, %{}, "Linus"}
...>   e -> e
...> end)
{:name, nil, [{:first, %{}, "Linus"}]}

iex> XmlBuilder.prewalk(
...>  [{:person, %{},
...>    [{:name, %{id: 123}, "Josh"},
...>     {:age, %{}, "21"}]}], fn
...>   {:age, _, _} -> {:age, %{}, "24"}
...>   e -> e
...> end)
...> |> hd()
...> |> XmlBuilder.generate(format: XmlBuilder.Format.None)
~s|<person><name id="123">Josh</name><age>24</age></person>|

  



  
    
      
      Link to this function
    
    prewalk(element, acc, pre)


      
       
       View Source
     


  


  

      

          @spec prewalk(
  XmlBuilder.Element.ast(),
  any(),
  (XmlBuilder.Element.as_tuple(), any() ->
     {XmlBuilder.Element.as_tuple(), any()})
) :: {XmlBuilder.Element.as_tuple(), any()}


      


Performs a depth-first, pre-order traversal of XML tree.
Returns a new tree of XmlBuilder.Element.t() where each element
  is the result of invoking fun on each corresponding element.

  
    
  
  Examples


iex> XmlBuilder.prewalk(XmlBuilder.element(:person), %{}, fn
...>   {x, nil, nil}, a -> {{x, %{}, []}, a}
...> end)
{{:person, %{}, []}, %{}}

iex> XmlBuilder.prewalk({:person, %{id: 1}, "Steve Jobs"}, %{}, fn
...>   {name, args, _content}, a -> {{name, args, "Linux Torvalds"}, a}
...> end)
{{:person, %{id: 1}, "Linux Torvalds"}, %{}}

iex> XmlBuilder.prewalk({:name, nil, [{:first, nil, "Steve"}]}, %{}, fn
...>   {:first, _, value}, a ->
...>     {{:first, %{}, "Linus"}, Map.put(a, :first, value)}
...>   e, a ->
...>     {e, a}
...> end)
{{:name, nil, [{:first, %{}, "Linus"}]}, %{first: "Steve"}}

iex> XmlBuilder.prewalk(
...>  [{:person, %{},
...>    [{:name, %{id: 123}, "Josh"},
...>     {:age, %{}, "21"}]}], %{}, fn
...>   {:age, _, _}, a -> {{:age, %{}, "24"}, a}
...>   e, a -> {e, a}
...> end)
...> |> elem(0)
...> |> hd()
...> |> XmlBuilder.generate(format: XmlBuilder.Format.None)
~s|<person><name id="123">Josh</name><age>24</age></person>|

  



  
    
      
      Link to this function
    
    traverse(element, acc, pre, post)


      
       
       View Source
     


  


  

      

          @spec traverse(
  XmlBuilder.Element.as_tuple(),
  any(),
  (XmlBuilder.Element.as_tuple(), any() ->
     {XmlBuilder.Element.as_tuple(), any()}),
  (XmlBuilder.Element.as_tuple(), any() ->
     {XmlBuilder.Element.as_tuple(), any()})
) :: {XmlBuilder.Element.as_tuple(), any()}


      


Performs a depth-first traversal of XML tree using an accumulator.

  


        

      



  

    
XmlBuilder.Element 
    



      
The internal representation of the XML element,
  responsible for parsing and normalization.
Generally, you won’t use this internal module.

      


      
        Summary


  
    Types
  


    
      
        as_tuple()

      


        The raw internal representation of the XML element



    


    
      
        ast()

      


        The AST representation of XML



    


    
      
        attrs()

      


        Attributes of the XML element



    


    
      
        content()

      


        Single inner subelement of the XML element



    


    
      
        contents()

      


        Whole inner content of the XML element



    


    
      
        name()

      


        Name of the XML element



    


    
      
        t()

      


        The internal representation of the XML element



    





      


      
        Types

        


  
    
      
      Link to this type
    
    as_tuple()


      
       
       View Source
     


  


  

      

          @type as_tuple() :: {name(), attrs(), contents()}


      


The raw internal representation of the XML element

  



  
    
      
      Link to this type
    
    ast()


      
       
       View Source
     


  


  

      

          @type ast() :: as_tuple() | [as_tuple()]


      


The AST representation of XML

  



  
    
      
      Link to this type
    
    attrs()


      
       
       View Source
     


  


  

      

          @type attrs() :: map() | nil


      


Attributes of the XML element

  



  
    
      
      Link to this type
    
    content()


      
       
       View Source
     


  


  

      

          @type content() :: tuple() | String.t()


      


Single inner subelement of the XML element

  



  
    
      
      Link to this type
    
    contents()


      
       
       View Source
     


  


  

      

          @type contents() :: [content()] | nil


      


Whole inner content of the XML element

  



  
    
      
      Link to this type
    
    name()


      
       
       View Source
     


  


  

      

          @type name() :: bitstring() | binary() | atom()


      


Name of the XML element

  



  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: %XmlBuilder.Element{attrs: attrs(), content: contents(), name: name()}


      


The internal representation of the XML element

  


        

      



  

    
XmlBuilder.Format behaviour
    



      
Base behaviour for all the formatters. There are two defauls formatters
  included: XmlBuilder.Format.None and XmlBuilder.Format.Indented.
To introduce the new formatter, one should use XmlBuilder.Format and
  possibly override indent/1 and/or intersperse/0 implementation(s):
defmodule MyCuteFormat do
  use XmlBuilder.Format, indent: "  ", intersperse: "\n"
end

      


      
        Summary


  
    Callbacks
  


    
      
        indent(non_neg_integer)

      


    


    
      
        intersperse()

      


    





      


      
        Callbacks

        


  
    
      
      Link to this callback
    
    indent(non_neg_integer)


      
       
       View Source
     


  


  

      

          @callback indent(non_neg_integer()) :: binary() | iodata()


      



  



  
    
      
      Link to this callback
    
    intersperse()


      
       
       View Source
     


  


  

      

          @callback intersperse() :: binary() | iodata()


      



  


        

      



  

    
XmlBuilder.Format.Indented 
    



      
Format for XmlBuilder.generate/2 to produce a human-readable indented XML.
Unlike XmlBuilder.Format.None, this formatter
  produces a human readable XML, indented with tabs and new lines, like this:
<person>
  <name id=\"123\">Josh</name>
  <age>21</age>
</person>
Normally you should not use this formatter explicitly, pass it as an
optional parameter to XmlBuilder.generate/2 instead.

      





  

    
XmlBuilder.Format.None 
    



      
Format for XmlBuilder.generate/2 to produce a machine-readable aka
  minified XML.
Unlike XmlBuilder.Format.Indented, this formatter does not
  produce a human readable XML, it simply spits out elements one immediately
after another, like this:
<person><name id=\"123\">Josh</name><age>21</age></person>
Normally you should not use this formatter explicitly, pass it as an
optional parameter to XmlBuilder.generate/2 instead.

      





  

    
XmlBuilder.Format.TabIndented 
    



      
Format for XmlBuilder.generate/2 to produce a human-readable indented XML.
Like XmlBuilder.Format.Indented, but indents the content with tabs instead of spaces.
Normally you should not use this formatter explicitly, pass it as an
optional parameter to XmlBuilder.generate/2 instead.

